

Pensieve header: January 18: Fibonacci Forever.

```
f1[0] = 1; f1[1] = 1;
f1[n_] := (f1[n] = f1[n - 1] + f1[n - 2])
```

```
Table[f1[k], {k, 1, 10}]
```

```
{1, 2, 3, 5, 8, 13, 21, 34, 55, 89}
```

```
f1[100]
```

```
573 147 844 013 817 084 101
```

```
f2[n_] := Module[{k = 1, prev, cur},
  prev = cur = 1;
  While[k < n, {prev, cur} = {cur, prev + cur}; ++k];
  cur
]
```

```
f2[10]
```

```
89
```

```
f2[100]
```

```
573 147 844 013 817 084 101
```

**? For**

For[start, test, incr, body] executes start, then repeatedly evaluates body and incr until test fails to give True. >>

```
f3[n_] := (For[k = cur = prev = 1, k < n, k += 1, {prev, cur} = {cur, prev + cur}]; cur)
```

```
f3[100]
```

```
573 147 844 013 817 084 101
```

**? Do**

Do[expr, n] evaluates expr n times.

Do[expr, {i, imax}] evaluates expr with the variable i successively taking on the values 1 through imax (in steps of 1).

Do[expr, {i, imin, imax}] starts with i = imin.

Do[expr, {i, imin, imax, di}] uses steps di.

Do[expr, {i, {i1, i2, ...}}] uses the successive values i1, i2, ...

Do[expr, {i, imin, imax}, {j, jmin, jmax}, ...] evaluates expr looping over different values of j etc. for each i. >>

```
f4[n_] := (
  prev = cur = 1;
  Do[{prev, cur} = {cur, prev + cur}, n - 1];
  cur)
```

**f4[100]**

573 147 844 013 817 084 101

**f4[2]**

2

```
f5[n_] := {1, 1, 1} //. {
  {n, _, c_} => c,
  {k_, p_, c_} => {k+1, c, p+c}
}
```

**f5[100]**

573 147 844 013 817 084 101

**Series** $\left[\frac{1}{1-x-x^2}, \{x, 0, 10\}\right]$

$1 + x + 2x^2 + 3x^3 + 5x^4 + 8x^5 + 13x^6 + 21x^7 + 34x^8 + 55x^9 + 89x^{10} + O[x]^{11}$

**Series** $\left[\frac{1}{1-x-x^2}, \{x, 0, 10\}\right]$  // **FullForm**

SeriesData[x, 0, List[1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89], 0, 11, 1]

**Series** $\left[\frac{1}{1-x-x^2}, \{x, 0, 100\}\right][[3]]$  // **Last**

573 147 844 013 817 084 101

**SeriesCoefficient** $\left[\frac{1}{1-x-x^2}, \{x, 0, 100\}\right]$

573 147 844 013 817 084 101

**D** $\left[\frac{1}{1-x-x^2}, \{x, 100\}\right]$   
 100!

573 147 844 013 817 084 101

**f8[n\_] := Sum** $[\text{Binomial}[n-k, k], \{k, 0, \text{Floor}[n/2]\}]$

**f8[100]**

573 147 844 013 817 084 101